

CENTRALESUPÉLEC

ARCHITECTURE DES ORDINATEURS REPORT

TD2

Designing a simple datapath in VHDL

Group :

Daniel STULBERG HUF
Lawson OLIVEIRA LIMA
Lucas VITORIANO DE QUEIROZ LIRA

Professors :

Salvador PEREA RUBEN
Hiet GUILLAUME



Contents

1	Introduction	1
2	Basic ALU design	1
3	Basic ALU + Output Register	2
4	Basic Datapath: ALU + Register File	4
5	Bonus question	7
6	Conclusion	11

1 Introduction

In this report, we will cover the process of simulate how a real hardware deal with instructions, registers, and arithmetic operations using VHDL. The decisions we made regarding the design of the architecture will be explained along with the source code of each step. All the simulations were ran using the Vivado software.

2 Basic ALU design

In this first step, the basic ALU with 4 instructions was built. The source code written in alu.vhd is shown below.

```

1  library ieee;
2  use ieee.std_logic_1164.all;
3  use ieee.numeric_std.all;
4
5  entity alu is
6      port(in1      : in  std_logic_vector(31 downto 0);
7           in2      : in  std_logic_vector(31 downto 0);
8           op       : in  std_logic_vector(1  downto 0);
9           res      : out std_logic_vector(31 downto 0));
10 end alu;
11
12 architecture behav of alu is
13 begin
14     process (in1, in2, op)
15     begin
16         case op is
17             when "00" =>
18                 res <= std_logic_vector(signed(in1) + signed(in2));
19             when "01" =>
20                 res <= std_logic_vector(signed(in1) - signed(in2));
21             when "10" =>
22                 res <= in1 and in2;
23             when "11" =>
24                 res <= in1 or in2;
25             when others =>
26                 null;
27         end case;
28     end process;
29 end;
```

Inside the architecture, the When statement was used to treat all the 2-bit input cases. For the operations of addition and subtraction, it was necessary to convert the inputs to “signed” in order to perform such arithmetic operations, and then reverse it back to “std_logic_vector” composite type to assign it to the output. As for the AND and OR operations, we directly used the corresponding logical operators between the two inputs and then assigned the result to the output. In other cases, the null statement will perform no action.

The result of the testbench simulation can be seen in the figure 1.




Figure 1: Simulation for the ALU benchmark.

3 Basic ALU + Output Register

In the second step, an output register was added after the ALU. The source code written in `alu_reg.vhd` is shown below.

```

1  library ieee;
2  use ieee.std_logic_1164.all;
3  use ieee.numeric_std.all;
4
5  entity alu_reg is
6      port(clk          : in  std_logic;
7            reset       : in  std_logic;
8            in1, in2    : in  std_logic_vector(31 downto 0);
9            op          : in  std_logic_vector(1 downto 0);
10           ena         : in  std_logic;
11           res         : out std_logic_vector(31 downto 0));
12 end alu_reg;
13
14 architecture behav of alu_reg is
15
16     signal alu_out      : std_logic_vector(31 downto 0);
17     signal alu_op       : std_logic_vector(1 downto 0);
18     signal RegWrite     : std_logic;
19
20     begin
21
22         -----
23         -- Control unit
24         -- + Generate the ALU control signal to select the required operation
25         --   / In a real CPU design this signal would result from decoding the instruction
26         --   / Suppose the ALU operation is already decoded and available in "Op" input
27         -- + Generate the write enable signal for the ALU result register
28         -----
29     process (clk)
30     begin
31         if rising_edge(clk) then
32             if (reset = '1') then
33                 alu_op      <= (others => '0');
34                 RegWrite    <= '0';
35             elsif (ena = '1') then
36                 -- Complete the code for alu_op and RegWrite
37                 alu_op <= op;
38                 RegWrite <= '1';
39             else
40                 -- Complete the code to make RegWrite active only 1 clock cycle
41                 RegWrite <= '0';
42             end if;
43         end if;
44     end process;
45
46     -----
47     -- ALU
48     -----
49     -- Use the ALU from before

```

```

50
51 process (in1, in2, alu_op)
52 begin
53     case alu_op is
54         when "00" =>
55             alu_out <= std_logic_vector(signed(in1) + signed(in2));
56         when "01" =>
57             alu_out <= std_logic_vector(signed(in1) - signed(in2));
58         when "10" =>
59             alu_out <= in1 and in2;
60         when "11" =>
61             alu_out <= in1 or in2;
62         when others =>
63             null;
64     end case;
65 end process;
66
67 -----
68 -- Output register to save ALU result
69 -----
70 -- Complete the code for output res
71 process (clk)
72 begin
73     if rising_edge(clk) then
74         if(RegWrite = '1') then
75             res <= alu_out;
76         elsif(reset = '1') then
77             res <= (others => '0');
78         end if;
79     end if;
80 end process;
81 end;

```

In this architecture, the design is split between control unit, ALU, and the output register. Moreover, the clock signal is being used as the design is now synchronous and sequential. For the control unit part, the process will only be activated once there is a rising edge in the clock counter. In the case where the reset is not activated and the ALU is enabled to receive signals, the variable `alu_op` will receive `op`, since we want the ALU to perform operations in this situation. `RegWrite` will also be activated because we want the results to be saved in the output register at this moment.

For the case when both reset and ALU are deactivated, we assigned the value '0' to `RegWrite` so that it will also become inactive (it will be activated only in the next clock cycle). As for the output register part, the output `res` will receive the signal from `alu_out` just when the clock is rising and the saving is enabled. Otherwise, if the clock is rising and the reset is activated, `res` will also be reset.

The result of the testbench simulation can be seen below




Figure 2: Simulation for the ALU+Output Register benchmark.

We can notice in the image that for a moment the output is undefined, this is because of the `alu_op`, as it is also undefined the alu doesn't have a operation to do, so his output is floating.

4 Basic Datapath: ALU + Register File

In the third step, a register file, instead of a unique output register, was added together with the ALU. The source code written in `alu_regfile.vhd` is shown below.

```

1  library ieee;
2  use ieee.std_logic_1164.all;
3  use ieee.numeric_std.all;
4
5  entity alu_regfile is
6      port(clk      : in  std_logic;
7            reset    : in  std_logic;
8            op       : in  std_logic_vector(1 downto 0);
9            Rreg1     : in  std_logic_vector(2 downto 0);
10           Rreg2     : in  std_logic_vector(2 downto 0);
11           Wreg      : in  std_logic_vector(2 downto 0);
12           ena       : in  std_logic;
13           res       : out std_logic_vector(31 downto 0);
14           init      : in  std_logic);
15  end alu_regfile;
16
17
18  architecture behav of alu_regfile is
19
20      -- ALU signals
21      signal alu_in1  : std_logic_vector(31 downto 0);
22      signal alu_in2  : std_logic_vector(31 downto 0);
23      signal alu_out   : std_logic_vector(31 downto 0);
24      signal alu_op    : std_logic_vector(1 downto 0);
25
26      -- Register file
27      type mem_array is array(7 downto 0) of std_logic_vector(31 downto 0);
28      signal reg_file: mem_array;
29      signal RegWrite : std_logic;
30
31  begin
32
33      -----
34      -- Control unit
35      -- + Generate the ALU control signal to select the required operation
36      --   / In a real CPU design this signal would result from decoding the instruction
37      --   / Suppose the ALU operation is already decoded and available in "op" input
38      -- + Generate the write signal to store the result in the register file
39      -----
40      -- Use the control unit from before
41      process (clk)
42      begin
43          if rising_edge(clk) then
44              if (reset = '1') then
45                  alu_op      <= (others => '0');
46                  RegWrite    <= '0';
47              elsif (ena = '1') then
48                  alu_op <= op;
49                  RegWrite <= '1';

```

```

50         else
51             RegWrite <= '0';
52         end if;
53     end if;
54 end process;
55
56 -----
57 -- Register file
58 -- Write: synchronous with the clock edge, write enable signal
59 -- Read: combinational outputs, no read enable signal
60 -----
61 process (clk)
62 begin
63     if rising_edge(clk) then
64         -- Complete the code for reg_file: check multi-ported memories on the slides
65         if (RegWrite = '1') then
66             reg_file(to_integer(unsigned(Wreg))) <= alu_out;
67         end if;
68         -- Dirty hack to initialize the register file for a simple, meaningful simulation.
69         -- So, only for simulation purposes in this special case.
70         -- This is NOT the way to do it for a real design
71         if (reset = '1') then
72             reg_file <= (0 => (others => '0'),
73                 1 => x"00000000",
74                 2 => x"00000000",
75                 3 => x"00000000",
76                 4 => x"00000000",
77                 5 => x"00000000",
78                 6 => x"00000000",
79                 7 => x"00000000",
80                 others => (others => '0'));
81
82             elsif (init = '1') then
83                 reg_file <= (0 => (others => '0'),
84                     1 => x"00000001",
85                     2 => x"00000002",
86                     3 => x"00000003",
87                     4 => x"00000004",
88                     5 => x"00000005",
89                     6 => x"00000006",
90                     7 => x"00000007",
91                     others => (others => '0'));
92             end if;
93         end if;
94     end process;
95
96 -- Complete the code
97 -- Register file outputs are the ALU inputs
98 alu_in1 <= reg_file(to_integer(unsigned(Rreg1)));
99 alu_in2 <= reg_file(to_integer(unsigned(Rreg2)));
100
101
102 -----

```


```

103  -- ALU
104  -----
105  -- Use the ALU from before
106  process (alu_op,alu_in1,alu_in2)
107  begin
108      case alu_op is
109          when "00" =>
110              alu_out <= std_logic_vector(signed(alu_in1) + signed(alu_in2));
111          when "01" =>
112              alu_out <= std_logic_vector(signed(alu_in1) - signed(alu_in2));
113          when "10" =>
114              alu_out <= alu_in1 and alu_in2;
115          when "11" =>
116              alu_out <= alu_in1 or alu_in2;
117          when others =>
118              null;
119          end case;
120      end process;
121
122  -----
123  -- Output
124  -----
125  -- Assign the output
126  res <= alu_out;
127  end behav;

```

In this architecture, the design is split between control unit, ALU, and the register file. For the write register signal in the register file, the ALU output will only be written in the desired register of the register file once there is a rising edge in the clock and when RegWrite is activated (which automatically implies that the enable signal is also activated). Wreg is the desired register in which data will be written, and is represented as an array of bits, and the register file can be seen as a matrix of 8 lines (registers) and 32 columns (array of bits). Since the register file expects one register in which data coming from the ALU output will be written, we need to convert the array of bits to its corresponding integer.

We also added the reset case inside the rising edge condition, which will erase all the content from the register file. For the register file outputs to be the ALU inputs, the same conversion operation done before is being done again in order to access and send the content from the register file (by the two read registers) to the ALU. Finally, the ALU output was assigned to the output port of the architecture. The result of the testbench simulation can be seen below.



5 Bonus question

Finally, for the bonus question, we have to read one instruction in the format RV32 and make the procedure respective to that function. The respective code to do that was written in `datapath_and_control_regfile.vhd` and is shown below.

```

1  library ieee;
2  use ieee.std_logic_1164.all;
3  use ieee.numeric_std.all;
4
5  entity datapath_and_control is
6      port(clk      : in  std_logic;
7            reset    : in  std_logic;
8            inst     : in  std_logic_vector(31 downto 0);
9            inst_we   : in  std_logic;
10           init     : in  std_logic);
11 end datapath_and_control;
12
13
14 architecture behav of datapath_and_control is
15     -- ALU signals
16     signal alu_in1  : std_logic_vector(31 downto 0);
17     signal alu_in2  : std_logic_vector(31 downto 0);
18     signal alu_out   : std_logic_vector(31 downto 0);
19     signal alu_op    : std_logic_vector(1 downto 0);
20
21     -- Register file
22     type mem_array is array(7 downto 0) of std_logic_vector(31 downto 0);
23     signal reg_file: mem_array;
24     -- Why size 3 ?
25     signal Rreg1    : std_logic_vector(2 downto 0);
26     signal Rreg2    : std_logic_vector(2 downto 0);
27     signal Wreg     : std_logic_vector(2 downto 0);
28
29     signal RegWrite : std_logic;
30
31     -- Instruction fields
32     signal IR       : std_logic_vector(31 downto 0);
33     signal funct7   : std_logic_vector(6 downto 0);
34     signal funct3   : std_logic_vector(2 downto 0);
35     signal rs1      : std_logic_vector(4 downto 0);
36     signal rs2      : std_logic_vector(4 downto 0);
37     signal rd       : std_logic_vector(4 downto 0);
38     signal opcode   : std_logic_vector(6 downto 0);
39
40     signal exec     : std_logic;
41
42 begin
43
44     -----
45     -- Instruction register
46     -----
47
48     -- Complete the code for the signals: IR and exec
49     -- IR: the instruction register

```

```

49  -- exec: we generate an auxiliary signal pulse, 1 clock cycle wide,
50  -- to enable the control unit
51
52
53  -----
54  -- Control unit
55  -- + 1. Generate the ALU control signal alu_op to select the required operation
56  --   / In a real CPU design this signal would result from decoding the instruction
57  --   / Suppose the ALU operation is already decoded and available in "op" input
58  -- + 2. Generate the write signal RegWrite to store the result in the register file
59  -- + 3. Extract the address of source and destination operands in the register file
60  -----
61
62
63  -- Let's make it a bit closer to RISC-V
64  -- Define RISC-V R-type instruction field names
65  funct7  <= IR(31 downto 25);
66  rs2     <= IR(24 downto 20);
67  rs1     <= IR(19 downto 15);
68  funct3  <= IR(14 downto 12);
69  rd      <= IR(11 downto 7);
70  opcode  <= IR(6  downto 0);
71
72  -- 1. and 2.
73  -- Complete the code for alu_op and RegWrite
74  process (clk)
75  begin
76      if rising_edge(clk) then
77          if (inst_we = '1') then
78              exec <= '1';
79              IR <= inst;
80          else
81              exec <= '0';
82          end if;
83      end if;
84  end process;
85
86  Rreg1  <= rs1(2 downto 0);
87  Rreg2  <= rs2(2 downto 0);
88  Wreg   <= rd(2 downto 0);
89
90  --Control Unit
91  process(clk)
92  begin
93      if rising_edge(clk) then
94          if (exec = '1') then
95              RegWrite <= '1';
96              case( funct7 ) is
97                  when "0100000" =>
98                      alu_op <= "01";
99                  when "0000000" =>
100                     case( funct3 ) is
101                         when "000" =>

```

```

102         alu_op <= "00";
103     when "110" =>
104         alu_op <= "10";
105     when "111" =>
106         alu_op <= "11";
107     when others =>
108         alu_op <= "00";
109     end case ;
110     when others =>
111         alu_op <= "00";
112     end case ;
113 else
114     RegWrite <= '0';
115     alu_op <= "00";
116 end if;
117 end if;
118 end process;
119 -- 3.
120 -- Complete the code for the source and destination operands
121 -- Since our regfile and datapath is a stripped down version,
122 -- we keep only the bits required to address the 8 registers of our register file
123
124
125 -----
126 -- Register file
127 -- Write: synchronous with the clock edge, write enable signal
128 -- Read: combinational outputs, no read enable signal
129 -----
130 -- Use the same register file as before...
131 alu_in1 <= reg_file(to_integer(unsigned(Rreg1)));
132 alu_in2 <= reg_file(to_integer(unsigned(Rreg2)));
133
134
135 process (clk)
136 begin
137     if rising_edge(clk) then
138         -- Complete the code for reg_file: check multi-ported memories on the slides
139         if(RegWrite = '1') then
140             reg_file(to_integer(unsigned(Wreg))) <= alu_out;
141         end if;
142         -- Dirty hack to initialize the register file for a simple, meaningful simulation.
143         -- So, only for simulation purposes in this special case.
144         -- This is NOT the way to do it for a real design
145         if (reset = '1') then
146             reg_file <= (0 => (others => '0'),
147                 1 => x"00000000",
148                 2 => x"00000000",
149                 3 => x"00000000",
150                 4 => x"00000000",
151                 5 => x"00000000",
152                 6 => x"00000000",
153                 7 => x"00000000",
154                 others => (others => '0'));

```

```

155
156     elsif(init = '1') then
157         reg_file <= (0 => (others => '0'),
158             1 => x"00000001",
159             2 => x"00000002",
160             3 => x"00000003",
161             4 => x"00000004",
162             5 => x"00000005",
163             6 => x"00000006",
164             7 => x"00000007",
165             others => (others => '0'));
166     end if;
167 end if;
168 end process;
169
170
171 -----
172 -- ALU
173 -----
174 -- Use the same ALU as before ...
175 process (alu_op,alu_in1,alu_in2)
176 begin
177     case alu_op is
178     when "00" =>
179         alu_out <= std_logic_vector(signed(alu_in1) + signed(alu_in2));
180     when "01" =>
181         alu_out <= std_logic_vector(signed(alu_in1) - signed(alu_in2));
182     when "10" =>
183         alu_out <= alu_in1 and alu_in2;
184     when "11" =>
185         alu_out <= alu_in1 or alu_in2;
186     when others =>
187         null;
188     end case;
189 end process;
190 end behav;

```

Inside the architecture, the design is split between the process responsible to deal with the instruction received, the control unit, ALU, and the register file. Because of the exec signal, the control unit will only send anything when the instruction is sent, exec will change only when we receive inst we. After the send of the exec signal, the control unit will change his outputs, which are responsible to control where the data will be written and which registers are necessary to perform the operations. For the register file, it will only change a register inside if RegWrite is 1 and, as the ALU isn't dependent of anything, it will always show a output.

We also added the reset case inside the rising edge condition, which will erase all the content from the register file. For the register file outputs to be the ALU inputs, the same conversion operation done before is being done again in order to access and send the content from the register file (by the two read registers) to the ALU. Finally, the ALU output was assigned to the wire of WriteRegister, to send the new value to be saved in the RegisterFile.

The result of the testbench simulation can be seen below.




Figure 4: Bonus question.

In this case the wires where the Instruction is separated in sub functions are undefined in the beginning of the simulation because they receive directly a chop of the instructions, and as this last is also undefined they can't be assigned to a value.

6 Conclusion

To conclude, all the operations in the three source codes are working as expected, and the simulation results are very similar to those presented in the instructions. In addition, we learn in this practical work how a processor works, from interpretation to task execution.